

When and how to set device parameters

Components such as PDFDevice are special external components also called output devices who's job it is to produce output from an Omnis report. Internal devices such as the Printer, Screen and Preview, and all external plug-in devices such as HTML, RTF and PDFDevice share a common Omnis notation interface. Part of this interface are configurable options called device parameters. For file based output devices these options typically include a destination file path and name.

External device parameters confusion

For external plug-in devices the only way to set device parameters is via the Omnis print destination window or the device notation `$devices.PDF.$setparam(param_constant,value)`. One important feature of device parameters is that the device parameter settings are saved by Omnis in the `omnis.cfg` file which means the next time you start Omnis they will contain the same values as you have last set them.

The first time PDFDevice is used to create output from a report, the file-name parameter will be empty and you will be presented with a put-file dialogue to specify a destination file-name. You may also be presented with this dialogue the first time you select the PDF destination in the print destination window. The file path and name that you select and enter will be stored in the Omnis config file and you will never again be prompted to select a destination file. This will now be your default destination file-name that is used whenever you print to the PDF destination, unless a new file-name is specified.

It is vitally important to set the intended file-name at the correct time. Confusion can occur if one implements code to manage the file selection/assignment if it is not done at the right time. This confusion is further exacerbated by a bug in some versions that causes an intermittent crash when no valid default file-name is stored in the `omnis.cfg` file prior to printing a report (see [Technical Note TN0019](#)). If a library is executed with a different installation of Omnis, i.e. a runtime which does not have the same `omnis.cfg` file, Omnis will cause PDFDevice to prompt the user with a put-file dialogue if the file-name device parameter is not set correctly prior to printing. It will also mean that other device options may not be applied correctly if they have not been set specifically.

When to set device parameters

Most device parameters must be set prior to PDFDevice receiving a request from Omnis to open the output file but it may not be entirely clear to many developers at which point during printing this request is sent. The request to open the output file is most certainly triggered when calling `$cdevice.$open()`, but when printing a report without specifically opening the device, the request is only triggered when the first report page is completed and send to the device by Omnis.

Therefore, when using `$cdevice.$open()`, device parameters must be set prior to calling `$open()`. Setting the file-name and other device parameters after calling `$cdevice.$open()` is futile as the PDF

file will be created using the default file-name and options stored in the omnis.cfg file during the call to \$open(). The notation \$device.\$open() and \$device.\$close() are typically only used when two or more reports are to be printed to the same PDF file.

But when not using \$device.\$open(), device parameters may be set just prior to the first page being completed by the Omnis print manager. This can be done in two places.

1. It can be done prior to calling ‘Set report name’ and ‘Print report’ or prior to the notation ‘\$reports.my_report.\$open()’. In this case you use \$device.\$setparam(kDevPdfFileName, file_path). Setting device parameters at this point will modify the parameters that are stored in the omnis.cfg file.
2. Device parameters can also be set in a method of your report prior to printing the first record. To do this you use the PDFDevice parameters report object as documented in the chapter ‘Designing PDFDevice’ and demonstrated in the examples.

Setting the device parameters here will NOT use or modify the parameters that are stored in the omnis.cfg file. However, if the file-name device parameter is empty in the omnis.cfg file, choosing this method of assigning device parameters will still cause Omnis to request PDFDevice to open the put-file selection dialogue as part of validating all parameters during the construction of the report instance, even though you may be changing the file-name a little later. Because of this behavior it is vital that you call \$device.\$setparam(kDevPdfFileName,...) to set a valid default file-name prior to printing any reports to prevent the user being presented with a put-file selection dialog because the omnis.cfg file does not contain a valid file-name. You must do this even though the default file-name may never be used.

The purpose of the device parameters report object

The device parameters report object was implemented for two reasons.

1. To cope with a multi-threaded web-client server environment. In such an environment it is vital that device parameters are calculated and assigned in a manner that does not cause conflict with other threads that may also be calculating and assigning device parameters.
2. To change some page related device options during printing, i.e. watermark or background image settings.

There is no need to use this object for any other reason.

Note: When using \$device.\$open() there is no point in using this report object except for altering page related options during printing.

Example code for setting the destination file-name.

The following example code demonstrates the correct setting of a destination file-name with user interaction and the printing of multiple reports to the same file.

```
;; prompt the user
Calculate theFileName as "untitled.pdf"
Do FileOps.$putfilename(theFileName) Returns #F
If flag false
    Quit method
End If
;; prepare PDF device for printing
Do $cdevice.$assign(kDevPdf)
Do $cdevice.$setparam(kDevPdfFileName,theFileName)
;; -- set any additional parameters, i.e. kDevPdfInfoAuthor
;; open the output file
Do $cdevice.$open()
;; print our reports to the same file
Set report name rReportOne
Print report
Set report name rReportTwo
Print report
;; finalize and close the output file
Do $cdevice.$close()
;; restore $cdevice
Do $cdevice.$assign(kDevScreen)
```